IE453 – Energy Systems Planning

Homework 1 - Report

Group 15

Members:

- Gökhan Akkaya
- Mürsel Başpınar
- Ali Barış Bilen
- Mustafa Kaan Çoban

First question:

a) By looking at the power curve of the turbine, we can calculate how much energy it can produce in one hour given a specific wind speed value. In an Excel spreadsheet, we calculated hourly energy generation that the turbine generates, and observed that there are many hours (almost all) in which hourly generation is lower than the demand (for instance, 8th of April between 00:00 and 01:00 the demand is 1,376 kWh, where hourly generation is 924 kWh). Having a storage system does not help because the turbine cannot even meet the current demand alone. This means, if we have only one wind turbine in the system, it is not possible to meet the hourly demand. Alternative ways to satisfy unmet demand include; installing a diesel generator that will help meeting the demand when the wind turbine falls behind or purchasing electricity from the grid (outsourcing).

b) Actual output (total energy generated) is 169,349 kW, capacity of the turbine is 1500 kW and there are total 168 hours in data sheet. Dividing actual output by (1500*168), we get **67.20%** as the capacity factor.

Total energy generated =	169.349,00	(kwh)
Capacity of turbine =	1500	(kw)
Total hours =	168	(hrs)
Capacity Factor =	67,20%	

c) Dividing average load (demand) by the peak load, (1360,08 kWh/1554 kWh), we get **87.52%** as the load factor. In the ideal case, the load factor should be 1.

Average demand =	1.360,08	(kWh)
Peak (max) load =	1.554,00	(kWh)
Load Factor =	87,52%	

Second question:

Our linear programming model to minimize the overall cost of electricity is the following;

$$\min \left(\frac{0.005}{(1-(1.005)^{-25\times 52})} \right) . \$2{,}000{,}000 \, . \, Turbine\_Count + \$0.35 . \sum_{t=1}^{168} Z_t$$

s.t.

Generation constraint: $$Energy\_generation_t \cdot Turbine\_count = Wind\_energy\_used_t + Energy\_stored_t + Energy\_curtailed_t \quad \forall t$$

Storage Constraint:
$$Storage\_level_t = Storage\_level_{t-1} + (0.88) . Energy\_stored_t - \frac{1}{(0.88)} Energy\_discharged_t \quad \forall t$$
$$Storage\_level_t \leq 500 \quad \forall t$$

Demand constraint: $$Hourly\_demand_t \leq Wind\_energy\_used_t + Energy\_discharged_t + Z_t \quad \forall t$$

$$Storage\_level_0 = 0 \quad \text{(Initially the storage is empty)}$$

$$1 \leq t \leq 168$$

All variables are nonnegative & Turbine_count is integer.

Where,

1) *Turbine_count* is the number of wind turbines to be installed
2) Z_t is the amount of electricity purchased from the grid at time t
3) $Energy\_generation_t$ is the amount of energy that is generated at time t
4) $Wind\_energy\_used_t$ is the amount of energy directly coming from the wind turbines at time t
5) $Energy\_stored_t$ is the amount of energy that is stored to the battery at time t
6) $Energy\_curtailed_t$ is the amount of energy that is not used (consumed) at time t

7) *Storage_level*$_t$ is the amount of energy stored in the battery at time t
8) *Energy_discharged*$_t$ is the amount of energy that is taken from the battery to fulfill the demand at time t
9) *Hourly_demand*$_t$ is the hourly demand
10) Model's base unit is kWh.

For the implementation of our model, we have wrote the following OPL code and solved it via CPLEX (The code files and the spreadsheet including our calculations are zipped and submitted along with this report. We put the code here only for convenience, to make the grader's life easier).

```
//Setting Parameters

 int Storage_capacity = ...;
 float Cost_of_grid_electricity = ...;
 float Storage_efficiency = ...;
 float Cost_of_a_single_turbine = ...;
 float Discount_factor_for_turbine = ...;

 range time = 1..168;
 range time4 = 0..168;

 int Energy_generation[time] = ...;
 int Hourly_demand[time] = ...;

// int Storage_level[time] = ...;

 //Setting decision variables

 dvar int Turbine_count;
 dvar float Grid_usage[time];
 dvar float Storage_level[time4];
 dvar float Energy_stored[time];
 dvar float Energy_discharged[time];
 dvar float Wind_energy_used[time];
 dvar float Energy_curtailed[time];

  //Objective function

 dexpr float Total_grid_cost = (sum(i in time)
Grid_usage[i])*Cost_of_grid_electricity;
 dexpr float Total_wind_turbine_cost =
(Turbine_count)*(Discount_factor_for_turbine)*(Cost_of_a_single_turbine);

 minimize Total_wind_turbine_cost + Total_grid_cost;
```

```
 //Constraints

 subject to{

   Turbine_count >= 0;

   forall (i in time)
       Grid_usage[i] >= 0;
   forall (i in time)
       Storage_level[i] >= 0;
   forall (i in time)
       Energy_stored[i] >= 0;
   forall (i in time)
       Energy_discharged[i] >= 0;
   forall (i in time)
       Wind_energy_used[i] >= 0;
   forall (i in time)
       Energy_curtailed[i] >= 0;

   forall (i in time)
     Generation_constraint:
       Energy_generation[i]*Turbine_count == Wind_energy_used[i] + Energy_stored[i] +
Energy_curtailed[i];


   Storage_level[0] == 0;

   forall (i in time)
     Storage_constraint1:
       Storage_level[i] == Storage_level[i-1] + (Storage_efficiency*Energy_stored[i])
- ((1/Storage_efficiency)*Energy_discharged[i]);

   forall (i in time)
     Storage_constraint2:
       Storage_level[i] <= Storage_capacity;

   forall (i in time)
     Demand_constraint:
       Hourly_demand[i] <= Wind_energy_used[i] + Energy_discharged[i] +
Grid_usage[i];
        }
```

Above is the *.mod* file. You may find our *.dat* file below.

```
//Scalar parameters

Storage_capacity = 500;
Cost_of_grid_electricity = 0.35;
Discount_factor_for_turbine = 0.0050076516;
Cost_of_a_single_turbine = 2000000;
Storage_efficiency = 0.88;
```

```
//Time series data

SheetConnection my_sheet("hw1-data.xlsx");

Energy_generation from SheetRead(my_sheet, "data_and_calculations!K2:K169");
Hourly_demand from SheetRead(my_sheet, "data_and_calculations!D2:D169");
```

Back to the question;

a) The optimal number of wind turbines to be installed is **2**.
Total cost of the system is **$28,930.495**.

b) Total demand satisfied from the grid is **25,428.25 kWh**. Total demand is **228,494 kWh**. So, 11.13% of the total demand is satisfied from the main grid.

c) Battery size is 0.5 MWh, which we enter as 500 kWh into our model. If the battery size was 1.5 GWh that would make 1500 kWh. We updated our *.dat* file accordingly and we get the following results;

- Total cost of electricity is decreased to **$27,837.687** from **$28,930.495**, which implies a savings of **$1092,808**.
- Total demand satisfied from the main grid is **22,305.94 kWh**. The percentage of total demand satisfied from the main grid dropped to **%9.76**.
- The optimal number of wind turbines to be installed is still **2**.

d) When we double the blade length, we quadruple the power output (and energy output too) because power output is directly proportional to the area swept by the blades as the hint suggests. Also the new wind turbine costs $2,500,000. When we solve the updated model, we obtain the following results;

- The optimal number of wind turbines to be installed is **1** in this case.
- Total cost of electricity is decreased to **$16,454.453**.

e) The cost of the storage system is 30 cents per Wh, which makes $300 per kWh. The discount factor for the turbine should be re-calculated since the lifetime of the storage is different than the wind turbine. Its discount factor is 0.0050280995. Using the wind turbine provided in d), we have updated our model as follows (updated/newly added parts are marked with red);

discount factor for storage $= 0.0050280995$

$$\min \left(\frac{0.005}{(1-(1.005)^{-25\times 52})}\right) \bullet \$2{,}500{,}000 \bullet \text{Turbine\_Count} + \$0.35 \cdot \sum_{t=1}^{168} z_t + \underbrace{df \cdot \$300}_{\$/\text{kWh}} \cdot \underbrace{\text{Storage\_cap}}_{\text{kWh}}$$

s.t.

Generation constraint: $\text{Energy\_generation}_t \cdot \text{Turbine\_count} = \text{Wind\_energy\_used}_t + \text{Energy\_stored}_t + \text{Energy\_curtailed}_t \quad \forall t$

Storage Constraint:

$$\text{Storage\_level}_t = \text{Storage\_level}_{t-1} + (0.88) \cdot \text{Energy\_stored}_t - \frac{1}{(0.88)} \text{Energy\_discharged}_t \quad \forall t$$

$$\text{Storage\_level}_t \leq \text{Storage\_cap} \quad \forall t$$

Demand constraint: $\text{Hourly\_demand}_t \leq \text{Wind\_energy\_used}_t + \text{Energy\_discharged}_t + z_t \quad \forall t$

$\text{Storage\_level}_0 = 0$ (Initially the storage is empty)

$1 \leq t \leq 168$

All variables are nonnegative & Turbine_count is integer.

OPL code:

```
//Setting Parameters

//int Storage_capacity = …; (not needed anymore)
float Cost_of_grid_electricity = ...;
float Storage_efficiency = ...;
float Cost_of_a_single_turbine = ...;
float Discount_factor_for_turbine = ...;
float Discount_factor_for_storage = ...;
float Cost_of_storage = ...;

range time = 1..168;
range time4 = 0..168;

int Energy_generation[time] = ...;
int Hourly_demand[time] = ...;

// int Storage_level[time] = ...;
```

```
 //Setting decision variables

 dvar int Turbine_count;
 dvar float Grid_usage[time];
 dvar float Storage_level[time4];
 dvar float Energy_stored[time];
 dvar float Energy_discharged[time];
 dvar float Wind_energy_used[time];
 dvar float Energy_curtailed[time];
 dvar int Storage_cap_dec_var;

  //Objective function

 dexpr float Total_grid_cost = (sum(i in time)
Grid_usage[i])*Cost_of_grid_electricity;
 dexpr float Total_wind_turbine_cost =
(Turbine_count)*(Discount_factor_for_turbine)*(Cost_of_a_single_turbine);
 dexpr float Total_storage_cost =
(Cost_of_storage)*(Discount_factor_for_storage)*(Storage_cap_dec_var);

 minimize Total_wind_turbine_cost + Total_grid_cost + Total_storage_cost;

 //Constraints

 subject to{

   Turbine_count >= 0;

   forall (i in time)
       Grid_usage[i] >= 0;
   forall (i in time)
       Storage_level[i] >= 0;
   forall (i in time)
       Energy_stored[i] >= 0;
   forall (i in time)
       Energy_discharged[i] >= 0;
   forall (i in time)
       Wind_energy_used[i] >= 0;
   forall (i in time)
       Energy_curtailed[i] >= 0;

   forall (i in time)
     Generation_constraint:
       Energy_generation[i]*Turbine_count == Wind_energy_used[i] + Energy_stored[i] +
Energy_curtailed[i];


   Storage_level[0] == 0;

   forall (i in time)
     Storage_constraint1:
       Storage_level[i] == Storage_level[i-1] + (Storage_efficiency*Energy_stored[i])
- ((1/Storage_efficiency)*Energy_discharged[i]);
```

```
    forall (i in time)
      Storage_constraint2:
        Storage_level[i] <= Storage_cap_dec_var;

    forall (i in time)
      Demand_constraint:
        Hourly_demand[i] <= Wind_energy_used[i] + Energy_discharged[i] +
Grid_usage[i];
     }
```

Above is the *.mod* file. You may find our updated *.dat* file below (updated/newly added parts are marked with red). Again, we also zipped these files for the second model and submitted along with the report.

```
//Scalar parameters

//Storage_capacity = 500; (not needed anymore)
Cost_of_grid_electricity = 0.35;
Discount_factor_for_turbine = 0.0050076516;
Discount_factor_for_storage = 0.0050280995;
Cost_of_storage = 300;
Cost_of_a_single_turbine = 2500000;
Storage_efficiency = 0.88;

//Time series data

SheetConnection my_sheet("hw1-data.xlsx");

Energy_generation from SheetRead(my_sheet, "data_and_calculations!L2:L169");
Hourly_demand from SheetRead(my_sheet, "data_and_calculations!D2:D169");
```

We have obtained the following results when we solve the updated model;

- Optimal size of the storage system is **1018**.
- Number of wind turbines is **1**.
- Total cost of the system is **$17,192.315.**

f) Capacity factors of wind turbines are as follows;

Capacity Factor Comparison				
Type	Total Energy Generated (kWh)	Capacity (kW)	Hours (hrs)	Cap. Factor
Wind turbine in part a)	169.349,00	1500	168	67,20%
Wind turbine in part c)	169.349,00	1500	168	67,20%
Wind turbine in part e)	677.396,00	6000	168	67,20%

Capacity factors are the same. When we quadruple the power output of the turbine, we also quadruple the actual output at the same time, so the capacity factor does not change.